



Security Audit

Report for LaunchPad Contracts

Date: Sep 14, 2024 **Version:** 1.0

Contact: contact@blocksec.com

Contents

Chapter 1 Introduction	1
1.1 About Target Contracts	1
1.2 Procedure of Auditing	2
1.2.1 Software Security	2
1.2.2 DeFi Security	2
1.2.3 NFT Security	3
1.2.4 Additional Recommendation	3
1.3 Security Model	3
Chapter 2 Findings	5
2.1 Software Security	6
2.1.1 Incorrect rounding direction of fee calculation in function <code>cancelOrder()</code>	6
2.1.2 Potential precision loss may prevent users from properly claiming	6
2.2 DeFi Security	8
2.2.1 Lack of updating <code>publicPhase.tokenPerSaleToken</code> in function <code>setPublicPhaseSaleCap()</code>	8
2.2.2 Lack of check in function <code>hasStarted()</code>	8
2.2.3 Potential failure of cancellation due to incorrect check in function <code>transferFundsToTreasury()</code>	11
2.2.4 Incorrect calculation in the function <code>quotePrice()</code>	11
2.2.5 Lack of check for user's remaining <code>publicPhaseDeposits</code> in function <code>cancelOrder()</code>	12
2.2.6 Lack of check in function <code>setPublicPhaseSaleCap()</code>	14
2.2.7 Lack of check on <code>_publicPhaseWithdrawalDuration</code> in function <code>configLaunchpad()</code>	14
2.2.8 Lack of check on <code>publicPhase.endTime</code> in function <code>startClaimingPhase()</code>	17
2.3 Additional Recommendation	18
2.3.1 Lack of check on <code>_startTime</code> in function <code>setPhase()</code>	18
2.3.2 Incorrect check in function <code>getCurrentPhaseInfo()</code>	19
2.4 Note	20
2.4.1 Potential centralization risks	20

Report Manifest

Item	Description
Client	Magpie
Target	LaunchPad Contracts

Version History

Version	Date	Description
1.0	Sep 14, 2024	First release

Signature

About BlockSec BlockSec focuses on the security of the blockchain ecosystem and collaborates with leading DeFi projects to secure their products. BlockSec is founded by top-notch security researchers and experienced experts from both academia and industry. They have published multiple blockchain security papers in prestigious conferences, reported several zero-day attacks of DeFi applications, and successfully protected digital assets that are worth more than 14 million dollars by blocking multiple attacks. They can be reached at [Email](#), [Twitter](#) and [Medium](#).

Chapter 1 Introduction

1.1 About Target Contracts

Information	Description
Type	Smart Contract
Language	Solidity
Approach	Semi-automatic and manual verification

The focus of this audit is on the LaunchPad Contracts¹ of the Magpie. The LaunchpadV2 contract is designed to facilitate token sales for new projects, offering a structured and secure process for both private and public phases. It ensures that token sales are conducted efficiently, with mechanisms in place for price discovery, vesting, and handling unsold quotas.

Please note that the audit scope is limited to the following smart contracts:

```
1 contracts/launchpad/LaunchPadV2.sol
2 contracts/launchpad/LaunchpadVestingV2.sol
```

Listing 1.1: Audit Scope for this Report

Other files are not within the scope of the audit. Additionally, all dependencies of the smart contracts within the audit scope are considered reliable in terms of both functionality and security, and are therefore not included in the audit scope.

The auditing process is iterative. Specifically, we would audit the commits that fix the discovered issues. If there are new issues, we will continue this process. The commit SHA values during the audit are shown in the following table. Our audit report is responsible for the code in the initial version ([Version 1](#)), as well as new code (in the following versions) to fix issues in the audit report.

Project	Version	Commit Hash
LaunchPad Contracts	Version 1	80fc03adcb81501c6f826ee3fac24670e81e9c88
	Version 2	40d0ecc80107f853da9aaaf0f0910d360bba1161

This audit report does not constitute investment advice or a personal recommendation. It does not consider, and should not be interpreted as considering or having any bearing on, the potential economics of a token, token sale or any other product, service or other asset. Any entity should not rely on this report in any way, including for the purpose of making any decisions to buy or sell any token, product, service or other asset.

This audit report is not an endorsement of any particular project or team, and the report does not guarantee the security of any particular project. This audit does not give any warranties on discovering all security issues of the smart contracts, i.e., the evaluation result does not guarantee the nonexistence of any further findings of security issues. As one audit cannot be considered comprehensive, we always recommend proceeding with independent audits and a public bug bounty program to ensure the security of smart contracts.

¹https://github.com/magpiexyz/magpie_contracts

The scope of this audit is limited to the code mentioned in Section 1.1. Unless explicitly specified, the security of the language itself (e.g., the solidity language), the underlying compiling toolchain and the computing infrastructure are out of the scope.

1.2 Procedure of Auditing

We perform the audit according to the following procedure.

- **Vulnerability Detection** We first scan smart contracts with automatic code analyzers, and then manually verify (reject or confirm) the issues reported by them.
- **Semantic Analysis** We study the business logic of smart contracts and conduct further investigation on the possible vulnerabilities using an automatic fuzzing tool (developed by our research team). We also manually analyze possible attack scenarios with independent auditors to cross-check the result.
- **Recommendation** We provide some useful advice to developers from the perspective of good programming practice, including gas optimization, code style, and etc.

We show the main concrete checkpoints in the following.

1.2.1 Software Security

- * Reentrancy
- * DoS
- * Access control
- * Data handling and data flow
- * Exception handling
- * Untrusted external call and control flow
- * Initialization consistency
- * Events operation
- * Error-prone randomness
- * Improper use of the proxy system

1.2.2 DeFi Security

- * Semantic consistency
- * Functionality consistency
- * Permission management
- * Business logic
- * Token operation
- * Emergency mechanism
- * Oracle security
- * Whitelist and blacklist
- * Economic impact
- * Batch transfer

1.2.3 NFT Security

- * Duplicated item
- * Verification of the token receiver
- * Off-chain metadata security

1.2.4 Additional Recommendation

- * Gas optimization
- * Code quality and style



Note The previous checkpoints are the main ones. We may use more checkpoints during the auditing process according to the functionality of the project.

1.3 Security Model

To evaluate the risk, we follow the standards or suggestions that are widely adopted by both industry and academy, including OWASP Risk Rating Methodology ² and Common Weakness Enumeration ³. The overall *severity* of the risk is determined by *likelihood* and *impact*. Specifically, likelihood is used to estimate how likely a particular vulnerability can be uncovered and exploited by an attacker, while impact is used to measure the consequences of a successful exploit.

In this report, both likelihood and impact are categorized into two ratings, i.e., *high* and *low* respectively, and their combinations are shown in Table 1.1.

Table 1.1: Vulnerability Severity Classification

Impact	High	High	Medium
	Low	Medium	Low
		High	Low
		Likelihood	

Accordingly, the severity measured in this report are classified into three categories: **High**, **Medium**, **Low**. For the sake of completeness, **Undetermined** is also used to cover circumstances when the risk cannot be well determined.

Furthermore, the status of a discovered item will fall into one of the following four categories:

- **Undetermined** No response yet.
- **Acknowledged** The item has been received by the client, but not confirmed yet.

²https://owasp.org/www-community/OWASP_Risk_Rating_Methodology

³<https://cwe.mitre.org/>

- **Confirmed** The item has been recognized by the client, but not fixed yet.
- **Fixed** The item has been confirmed and fixed by the client.

Chapter 2 Findings

In total, we found **ten** potential security issues. Besides, we have **two** recommendations and **one** note.

- High Risk: 0
- Medium Risk: 3
- Low Risk: 7
- Recommendation: 2
- Note: 1

ID	Severity	Description	Category	Status
1	Low	Incorrect rounding direction of fee calculation in function <code>cancelOrder()</code>	Software Security	Fixed
2	Low	Potential precision loss may prevent users from properly claiming	Software Security	Fixed
3	Medium	Lack of updating <code>publicPhase.tokenPerSaleToken</code> in function <code>setPublicPhaseSaleCap()</code>	DeFi Security	Confirmed
4	Medium	Lack of check in function <code>hasStarted()</code>	DeFi Security	Fixed
5	Low	Potential failure of cancellation due to incorrect check in function <code>transferFundsToTreasury()</code>	DeFi Security	Fixed
6	Low	Incorrect calculation in the function <code>quotePrice()</code>	DeFi Security	Fixed
7	Low	Lack of check for user's remaining <code>publicPhaseDeposits</code> in function <code>cancelOrder()</code>	DeFi Security	Fixed
8	Low	Lack of check in function <code>setPublicPhaseSaleCap()</code>	DeFi Security	Fixed
9	Medium	Lack of check on <code>_publicPhaseWithdrawalDuration</code> in function <code>configLaunchpad()</code>	DeFi Security	Confirmed
10	Low	Lack of check on <code>publicPhase.endTime</code> in function <code>startClaimingPhase()</code>	DeFi Security	Fixed
11	-	Lack of check on <code>_startTime</code> in function <code>setPhase()</code>	Recommendation	Fixed
12	-	Incorrect check in function <code>getCurrentPhaseInfo()</code>	Recommendation	Fixed
13	-	Potential centralization risks	Note	-

The details are provided in the following sections.

2.1 Software Security

2.1.1 Incorrect rounding direction of fee calculation in function `cancelOrder()`

Severity Low

Status Fixed in [Version 2](#)

Introduced by [Version 1](#)

Description In the `cancelOrder()` function, a `cancellationFee` is charged, and in the formula for calculating the fee at line 320, rounding down is used. Users can bypass the fee collection with a rather small amount. The fee calculation should instead be rounded up.

```

306  function cancelOrder(uint256 _amountToRefund) external whenNotPaused isSaleActive nonReentrant
      {
307      if (_amountToRefund == 0) revert InvalidAmount();
308
309      (bool isPrivatePhase, ) = getCurrentPhaseInfo();
310      if (isPrivatePhase) revert InvalidPhase();
311
312      uint256 withdrawalPeriodEnd = publicPhase.startTime + publicPhaseWithdrawalDuration;
313
314      if (block.timestamp > withdrawalPeriodEnd) revert PublicPhaseWithdrawalPeriodOver();
315
316      UserInfo storage user = userInfo[msg.sender];
317      if (_amountToRefund > user.publicPhaseDeposits) revert WithdrawExceedsDeposit();
318
319      //charge a fee on cancellation
320      uint256 fee = (_amountToRefund * cancellationFee) / DENOMINATOR;
321      uint256 amountAfterFee = _amountToRefund - fee;
322      accumulatedFees += fee;
323
324      user.publicPhaseDeposits -= _amountToRefund;
325      totalRaised -= _amountToRefund;
326      publicPhase.saleTokenDeposits -= _amountToRefund;
327      _rebalanceAndUpdate();
328      IERC20(saleToken).safeTransfer(msg.sender, amountAfterFee);
329      emit OrderCancelled(msg.sender, _amountToRefund, fee);
330  }

```

Listing 2.1: `contracts/launchpad/LaunchPadV2.sol`

Impact Users can bypass the `cancellationFee` collection.

Suggestion Use rounding up in the fee calculation.

2.1.2 Potential precision loss may prevent users from properly claiming

Severity Low

Status Fixed in [Version 2](#)

Introduced by [Version 1](#)

Description In the `_checkValidCapAndUpdate()` function, `allocatedInPrivatePhase` is used to record the number of project tokens allocated to the user during the private phase. In the calculation of `publicPhase.tokenPerSaleToken`, `publicPhase.saleCap` is reduced by `allocatedInPrivatePhase` to calculate the `tokenPerSaleToken`.

If a user purchases project tokens multiple times during the private phase, `allocatedInPrivatePhase` may record a value lower than the total amount the user ultimately claims due to precision loss accumulation (e.g., $3/2 + 3/2 = 2$ while $(3+3)/2 = 3$). This results in the calculated `publicPhase.tokenPerSaleToken` being slightly higher, which can cause the total number of tokens the user can claim during the final claim process to exceed the `publicPhase.saleCap`.

```
547 function _checkValidCapAndUpdate(uint256 _saleTokenAmount) internal {
548     uint256 _toAllocate = _tokenAllocBySale(_saleTokenAmount, privatePhase.tokenPerSaleToken);
549     if (_toAllocate == 0) revert ZeroAllocation();
550
551
552     allocatedInPrivatePhase += _toAllocate;
553     if (allocatedInPrivatePhase > privatePhase.saleCap) revert NotEnoughToken();
554
555
556     uint256 privatePhasePurchased = getUserPurchasedProjectTokens(msg.sender, true);
557     uint256 _userCap = (userInfo[msg.sender].priorityQuota * privatePhase.priorityMultiplier) /
558         DENOMINATOR;
559     if (privatePhasePurchased + _toAllocate > _userCap) revert ExceedsUserPriorityCap();
560 }
```

Listing 2.2: contracts/launchpad/LaunchPadV2.sol

```
561 function _tokenAllocBySale(
562     uint256 _saleTokenAmount,
563     uint256 _tokenPerSaleToken
564 ) internal view returns (uint256) {
565     uint256 numerator = _saleTokenAmount * _tokenPerSaleToken * 10 ** projectTokenDecimals;
566     uint256 denominator = DENOMINATOR * 10 ** saleTokenDecimals;
567     return numerator / denominator;
568 }
```

Listing 2.3: contracts/launchpad/LaunchPadV2.sol

```
593 function _getRebalancedTokenPerSaleToken(
594     uint256 _saleTokenDeposits
595 ) internal view returns (uint256) {
596     if (_saleTokenDeposits == 0) {
597         return publicPhaseMaxTokenPerSale;
598     } else {
599         uint256 rebalancedTokenPerSaleToken = ((publicPhase.saleCap - allocatedInPrivatePhase) *
600             (10 ** saleTokenDecimals) *
601             DENOMINATOR) / (_saleTokenDeposits * (10 ** projectTokenDecimals));
602         return
603             rebalancedTokenPerSaleToken < publicPhaseMaxTokenPerSale
604                 ? rebalancedTokenPerSaleToken
605                 : publicPhaseMaxTokenPerSale;
606     }
607 }
```

```
606     }
607 }
```

Listing 2.4: contracts/launchpad/LaunchPadV2.sol

Impact The last user may fail to claim due to insufficient token balance.

Suggestion Use rounding up in the `allocatedInPrivatePhase` calculation.

2.2 DeFi Security

2.2.1 Lack of updating `publicPhase.tokenPerSaleToken` in function

`setPublicPhaseSaleCap()`

Severity Medium

Status Confirmed

Introduced by Version 1

Description When the protocol's `owner` invokes function `setPublicPhaseSaleCap()` to reset the `publicPhase.saleCap`, the function does not invoke function `_rebalanceAndUpdate()` to update the price of the projectToken. This can result in the price becoming outdated, which may lead to losses for users.

```
513 function setPublicPhaseSaleCap(uint256 _saleCap) external onlyOwner {
514     if (privatePhase.saleCap != 0 && _saleCap < privatePhase.saleCap) revert InvalidSaleCap();
515
516
517     emit PhaseSaleCapUpdated(publicPhase.saleCap, _saleCap);
518     publicPhase.saleCap = _saleCap;
519 }
```

Listing 2.5: contracts/launchpad/LaunchPadV2.sol

Impact The function `setPublicPhaseSaleCap()` does not update the price in a timely manner, which may result in user losses.

Suggestion Timely invoking function `_rebalanceAndUpdate()` after `publicPhase.saleCap` is set to the new value.

Feedback from the Project Correct, but we will only invoke this function only once a few minutes before the public phase starts. After the public phase has started this function will never be invoked.

2.2.2 Lack of check in function `hasStarted()`

Severity Medium

Status Fixed in Version 2

Introduced by Version 1

Description The protocol's `owner` can set `privatePhase` or `publicPhase` using the function `setPhase()`. Specifically, there is no required order for the `owner` to set the `privatePhase` and

`publicPhase`, but the `privatePhase` must start first. If the `owner` sets the `publicPhase` without setting the `privatePhase`, users can invoke the function `buy()` to purchase `projectToken` as the function `hasStarted()` does not check whether `privatePhase.startTime` is 0. This allows users to skip the `privatePhase` and directly enter the `publicPhase`.

```
379 function setPhase(
380     uint32 _startTime,
381     uint32 _endTime,
382     uint256 _saleCap,
383     uint256 _tokenPerSaleToken,
384     uint256 _priorityMultiplier,
385     bool _isPrivate,
386     uint256 _cliffDuration
387 ) external onlyBeforeSale onlyOwner {
388     if (_startTime == 0 || _endTime <= _startTime || _endTime <= block.timestamp)
389         revert InvalidTime();
390     if (
391         _tokenPerSaleToken <= 0 ||
392         (_isPrivate &&
393             publicPhaseMaxTokenPerSale != 0 &&
394             publicPhaseMaxTokenPerSale > _tokenPerSaleToken)
395     ) revert InvalidPerSaleAmount();
396
397     PhaseInfo storage phase = _isPrivate ? privatePhase : publicPhase;
398
399
400
401     if (
402         (_isPrivate && publicPhase.endTime != 0 && _endTime != publicPhase.startTime) ||
403         (!_isPrivate && privatePhase.endTime != 0 && _startTime != privatePhase.endTime)
404     ) revert InvalidTime();
405
406
407     if (
408         (_isPrivate && publicPhase.saleCap != 0 && _saleCap > publicPhase.saleCap) ||
409         (!_isPrivate && privatePhase.saleCap != 0 && _saleCap < privatePhase.saleCap)
410     ) revert InvalidSaleCap();
411
412
413     phase.startTime = _startTime;
414     phase.endTime = _endTime;
415     phase.saleCap = _saleCap;
416     phase.tokenPerSaleToken = _tokenPerSaleToken;
417     phase.priorityMultiplier = _priorityMultiplier;
418     phase.cliffDuration = _cliffDuration;
419
420
421     emit PhaseUpdated(
422         _startTime,
423         _endTime,
424         _saleCap,
425         _tokenPerSaleToken,
426         _priorityMultiplier,
```

```
427     _cliffDuration
428   );
429 }
```

Listing 2.6: contracts/launchpad/LaunchPadV2.sol

```
278 function buy(uint256 _amount) external whenNotPaused isSaleActive nonReentrant {
279     if (_amount < min_sale_token_amount) {
280         revert InvalidAmount();
281     }
282
283
284     (bool isPrivatePhase, ) = getCurrentPhaseInfo();
285
286
287     PhaseInfo storage phaseInfo = isPrivatePhase ? privatePhase : publicPhase;
288
289
290     totalRaised += _amount;
291     phaseInfo.saleTokenDeposits += _amount;
292     UserInfo storage user = userInfo[msg.sender];
293
294
295     if (isPrivatePhase) {
296         _checkValidCapAndUpdate(_amount);
297         user.privatePhaseDeposits += _amount;
298     } else if (phaseInfo.saleTokenDeposits > publicPhaseDepositCap) {
299         revert PublicPhaseDepositCapExceeded();
300     } else {
301         user.publicPhaseDeposits += _amount;
302         _rebalanceAndUpdate();
303     }
304
305
306     IERC20(saleToken).safeTransferFrom(msg.sender, address(this), _amount);
307     emit AllocationPurchased(msg.sender, _amount);
308 }
```

Listing 2.7: contracts/launchpad/LaunchPadV2.sol

```
151 modifier isSaleActive() {
152     if (!hasStarted()) revert SaleNotStarted();
153     if (hasEnded()) revert SaleCompleted();
154     _;
155 }
```

Listing 2.8: contracts/launchpad/LaunchPadV2.sol

```
174 function hasStarted() public view returns (bool) {
175     return block.timestamp >= privatePhase.startTime;
176 }
```

Listing 2.9: contracts/launchpad/LaunchPadV2.sol

Impact The protocol may skip the privatePhase and directly enter the publicPhase.

Suggestion Add a check in the function `hasStarted()` to ensure that `privatePhase.startTime` is not equal to 0.

2.2.3 Potential failure of cancellation due to incorrect check in function

`transferFundsToTreasury()`

Severity Low

Status Fixed in [Version 2](#)

Introduced by [Version 1](#)

Description Function `transferFundsToTreasury()` allows the privileged owner to transfer deposited sale tokens to `treasury` when the current `block.timestamp` reaches the specified `withdrawalPeriodEnd`. This includes the case where `block.timestamp` is exactly equal to `withdrawalPeriodEnd`. However, users are also allowed to cancel previous purchases at this moment through the function `cancelOrder()`. In this case, if the owner transfers the sale token out first, the user will not be able to cancel the order, which is against the design.

```

497 function transferFundsToTreasury(uint256 _amount) external onlyOwner {
498     uint256 withdrawalPeriodEnd = publicPhase.startTime + publicPhaseWithdrawalDuration;
499     if (block.timestamp >= publicPhase.startTime && block.timestamp < withdrawalPeriodEnd)
500         revert TransferNotAllowed();
501
502
503     if (IERC20(saleToken).balanceOf(address(this)) < _amount) revert InvalidAmount();
504     IERC20(saleToken).safeTransfer(treasury, _amount);
505     emit TransferredToTreasury(saleToken, _amount);
506 }

```

Listing 2.10: contracts/launchpad/LaunchPadV2.sol

Impact Users can not cancel at the time of `withdrawalPeriodEnd`.

Suggestion Revise the check in function `transferFundsToTreasury()` to ensure the owner can only withdraw after `withdrawalPeriodEnd`.

2.2.4 Incorrect calculation in the function `quotePrice()`

Severity Low

Status Fixed in [Version 2](#)

Introduced by [Version 1](#)

Description The function `quotePrice()` returns the price of `projectToken` relative to `saleToken`. Specifically, the function incorrectly assumes that the decimal value of `saleToken` is `1e18` in the calculation. However, the `decimals` of `saleToken` (e.g., USDC) may not actually be `1e18`. This will ultimately result in `quoteToken` returning an incorrect price.

```

256 function quotePrice(
257     uint256 _amount,
258     bool _isBuy

```

```

259 ) external view whenNotPaused isSaleActive returns (uint256) {
260     (bool isPrivatePhase, PhaseInfo memory phaseInfo) = getCurrentPhaseInfo();
261
262
263     if (_amount < min_sale_token_amount || (!_isBuy && phaseInfo.saleTokenDeposits < _amount)) {
264         revert InvalidAmount();
265     }
266
267
268     if (!isPrivatePhase) {
269         uint256 rebalancedTokenPerSaleToken = _getRebalancedTokenPerSaleToken(
270             _isBuy
271             ? phaseInfo.saleTokenDeposits + _amount
272             : phaseInfo.saleTokenDeposits - _amount
273         );
274         phaseInfo.tokenPerSaleToken = rebalancedTokenPerSaleToken;
275     }
276     return ((DENOMINATOR * 1 ether) / phaseInfo.tokenPerSaleToken);
277 }

```

Listing 2.11: contracts/launchpad/LaunchPadV2.sol

Impact Function `quotePrice()` will return an incorrect price.

Suggestion Change `(DENOMINATOR * 1 ether) / phaseInfo.tokenPerSaleToken` to `(DENOMINATOR * 10**IERC20(saleToken).decimals()) / phaseInfo.tokenPerSaleToken`.

2.2.5 Lack of check for user's remaining `publicPhaseDeposits` in function `cancelOrder()`

Severity Low

Status Fixed in [Version 2](#)

Introduced by [Version 1](#)

Description Before the end of the `publicPhase`, users can cancel their orders via the function `cancelOrder()`. However, there is no check to ensure that the user's remaining `publicPhaseDeposits` is greater than or equal to `min_sale_token_amount`. Specifically, this is inconsistent with the check in the function `buy()`.

```

278 function buy(uint256 _amount) external whenNotPaused isSaleActive nonReentrant {
279     if (_amount < min_sale_token_amount) {
280         revert InvalidAmount();
281     }
282
283
284     (bool isPrivatePhase, ) = getCurrentPhaseInfo();
285
286
287     PhaseInfo storage phaseInfo = isPrivatePhase ? privatePhase : publicPhase;
288
289
290     totalRaised += _amount;

```

```
291     phaseInfo.saleTokenDeposits += _amount;
292     UserInfo storage user = userInfo[msg.sender];
293
294
295     if (isPrivatePhase) {
296         _checkValidCapAndUpdate(_amount);
297         user.privatePhaseDeposits += _amount;
298     } else if (phaseInfo.saleTokenDeposits > publicPhaseDepositCap) {
299         revert PublicPhaseDepositCapExceeded();
300     } else {
301         user.publicPhaseDeposits += _amount;
302         _rebalanceAndUpdate();
303     }
304
305
306     IERC20(saleToken).safeTransferFrom(msg.sender, address(this), _amount);
307     emit AllocationPurchased(msg.sender, _amount);
308 }
```

Listing 2.12: contracts/launchpad/LaunchPadV2.sol

```
306     function cancelOrder(uint256 _amountToRefund) external whenNotPaused isSaleActive nonReentrant
307     {
308         if (_amountToRefund == 0) revert InvalidAmount();
309
310         (bool isPrivatePhase, ) = getCurrentPhaseInfo();
311         if (isPrivatePhase) revert InvalidPhase();
312
313
314         uint256 withdrawalPeriodEnd = publicPhase.startTime + publicPhaseWithdrawalDuration;
315
316
317         if (block.timestamp > withdrawalPeriodEnd) revert PublicPhaseWithdrawalPeriodOver();
318
319
320         UserInfo storage user = userInfo[msg.sender];
321         if (_amountToRefund > user.publicPhaseDeposits) revert WithdrawExceedsDeposit();
322
323
324         //charge a fee on cancellation
325         uint256 fee = (_amountToRefund * cancellationFee) / DENOMINATOR;
326         uint256 amountAfterFee = _amountToRefund - fee;
327         accumulatedFees += fee;
328
329
330         user.publicPhaseDeposits -= _amountToRefund;
331         totalRaised -= _amountToRefund;
332         publicPhase.saleTokenDeposits -= _amountToRefund;
333         _rebalanceAndUpdate();
334         IERC20(saleToken).safeTransfer(msg.sender, amountAfterFee);
335         emit OrderCancelled(msg.sender, _amountToRefund, fee);
336     }
```


Listing 2.13: contracts/launchpad/LaunchPadV2.sol

Impact The user's remaining `publicPhaseDeposits` may be less than `min_sale_token_amount`.

Suggestion Add a check to ensure that the user's remaining `publicPhaseDeposits` is greater than or equal to `min_sale_token_amount`.

2.2.6 Lack of check in function `setPublicPhaseSaleCap()`

Severity Low

Status Fixed in [Version 2](#)

Introduced by [Version 1](#)

Description The protocol's `owner` can set the `publicPhase.saleCap` through the function `setPublicPhaseSaleCap()`. Specifically, `publicPhase.saleCap` affects the price of `projectToken` during the `publicPhase`. Thus, the function `setPublicPhaseSaleCap()` can only be invoked before `publicPhase.endTime` to ensure that the price of `projectToken` does not change once the `publicPhase` has ended.

```

513 function setPublicPhaseSaleCap(uint256 _saleCap) external onlyOwner {
514     if (privatePhase.saleCap != 0 && _saleCap < privatePhase.saleCap) revert InvalidSaleCap();
515
516
517     emit PhaseSaleCapUpdated(publicPhase.saleCap, _saleCap);
518     publicPhase.saleCap = _saleCap;
519 }

```

Listing 2.14: contracts/launchpad/LaunchPadV2.sol

Impact The price of `projectToken` may still change after the `publicPhase` ends.

Suggestion Add a check to ensure that `setPublicPhaseSaleCap` can only be invoked before the end of the `publicPhase`.

2.2.7 Lack of check on `_publicPhaseWithdrawalDuration` in function `configLaunchpad()`

Severity Medium

Status Confirmed

Introduced by [Version 1](#)

Description Users can close their orders via the function `cancelOrder()` before `publicPhase.startTime + publicPhaseWithdrawalDuration`. However, after the `publicPhase` begins, a malicious user can deposit a large amount of `saleToken` to reach the `publicPhase` sale cap, preventing other users from making purchases. If the `publicPhase.endTime` is close to or the same as `publicPhase.startTime + publicPhaseWithdrawalDuration`, the malicious user can withdraw part of the `saleToken` in the last allowed time window by canceling orders.

In this case, the malicious user can arbitrarily control the price of the `projectToken`, disrupting the entire token sale process and preventing other users from purchasing normally.

```
427 function configLaunchpad(
428     address _projectToken,
429     address _saleToken,
430     address _vestingContract,
431     address _treasury,
432     uint256 _privatePhaseVestingPart,
433     uint256 _publicPhaseVestingPart,
434     uint256 _minSaleTokenAmount,
435     uint32 _publicPhaseWithdrawalDuration,
436     uint256 _publicPhaseDepositCap,
437     uint256 _publicPhaseMaxTokenPerSale,
438     uint256 _cancellationFee
439 ) public onlyBeforeSale onlyOwner {
440     if (
441         _treasury == address(0) ||
442         _projectToken == address(0) ||
443         _saleToken == address(0) ||
444         _vestingContract == address(0)
445     ) revert ZeroAddress();
446     if (_privatePhaseVestingPart >= DENOMINATOR || _publicPhaseVestingPart >= DENOMINATOR)
447         revert InvalidFDVPart();
448
449
450     if (
451         _publicPhaseMaxTokenPerSale <= 0 ||
452         (privatePhase.tokenPerSaleToken != 0 &&
453             _publicPhaseMaxTokenPerSale > privatePhase.tokenPerSaleToken)
454     ) revert InvalidPerSaleAmount();
455
456
457     uint8 tempProjectTokenDecimals = IERC20Metadata(_projectToken).decimals();
458     uint8 tempSaleTokenDecimals = IERC20Metadata(_saleToken).decimals();
459     if (tempSaleTokenDecimals > tempProjectTokenDecimals) revert TokenDecimalExceedsLimit();
460
461
462     if (_cancellationFee > DENOMINATOR) revert InvalidFeeAmount();
463
464
465     projectToken = _projectToken;
466     saleToken = _saleToken;
467     projectTokenDecimals = tempProjectTokenDecimals;
468     saleTokenDecimals = tempSaleTokenDecimals;
469     vestingContract = ILaunchpadVesting(_vestingContract);
470     treasury = _treasury;
471     PRIVATE_PHASE_VESTING_PART = _privatePhaseVestingPart;
472     PUBLIC_PHASE_VESTING_PART = _publicPhaseVestingPart;
473     min_sale_token_amount = _minSaleTokenAmount;
474     publicPhaseWithdrawalDuration = _publicPhaseWithdrawalDuration;
475     publicPhaseDepositCap = _publicPhaseDepositCap;
476     publicPhaseMaxTokenPerSale = _publicPhaseMaxTokenPerSale;
```

```
477     cancellationFee = _cancellationFee;
478
479
480     emit LaunchpadConfigured(
481         _projectToken,
482         _saleToken,
483         _vestingContract,
484         _treasury,
485         _privatePhaseVestingPart,
486         _publicPhaseVestingPart,
487         _minSaleTokenAmount,
488         _publicPhaseWithdrawalDuration,
489         _publicPhaseDepositCap,
490         _publicPhaseMaxTokenPerSale,
491         _cancellationFee
492     );
493 }
```

Listing 2.15: contracts/launchpad/LaunchPadV2.sol

```
306     function cancelOrder(uint256 _amountToRefund) external whenNotPaused isSaleActive nonReentrant
307     {
308         if (_amountToRefund == 0) revert InvalidAmount();
309
310         (bool isPrivatePhase, ) = getCurrentPhaseInfo();
311         if (isPrivatePhase) revert InvalidPhase();
312
313
314         uint256 withdrawalPeriodEnd = publicPhase.startTime + publicPhaseWithdrawalDuration;
315
316
317         if (block.timestamp > withdrawalPeriodEnd) revert PublicPhaseWithdrawalPeriodOver();
318
319
320         UserInfo storage user = userInfo[msg.sender];
321         if (_amountToRefund > user.publicPhaseDeposits) revert WithdrawExceedsDeposit();
322
323
324         //charge a fee on cancellation
325         uint256 fee = (_amountToRefund * cancellationFee) / DENOMINATOR;
326         uint256 amountAfterFee = _amountToRefund - fee;
327         accumulatedFees += fee;
328
329
330         user.publicPhaseDeposits -= _amountToRefund;
331         totalRaised -= _amountToRefund;
332         publicPhase.saleTokenDeposits -= _amountToRefund;
333         _rebalanceAndUpdate();
334         IERC20(saleToken).safeTransfer(msg.sender, amountAfterFee);
335         emit OrderCancelled(msg.sender, _amountToRefund, fee);
336     }
```

Listing 2.16: contracts/launchpad/LaunchPadV2.sol

Impact A malicious user may manipulate the price of the `projectToken`, affecting the ability of other users to make purchases.

Suggestion Add a check to ensure there is sufficient time between the time at last allowable cancel order and `publicPhase.endTime`.

Feedback from the Project Noted, we will take care of this while doing configurations

2.2.8 Lack of check on `publicPhase.endTime` in function `startClaimingPhase()`

Severity Low

Status Fixed in [Version 2](#)

Introduced by [Version 1](#)

Description The function `startClaimingPhase()` is used to start the claiming phase after the sales have ended. However, the check uses `!hasEnded()` to determine if the sales have ended, which is incorrect.

Specifically, the `hasEnded()` function only checks whether the current `block.timestamp` is greater than the end time of the `public phase` while ignoring the possibility that the public phase may not have been set (`endTime = 0`), which is incorrect.

```
365  /// @dev Start Tokens Claiming Phase
366  function startClaimingPhase() external onlyOwner {
367      if (!hasEnded()) revert SaleNotCompleted();
368      if (canClaimTokens) revert ClaimingPhaseAlreadyStarted();
369
370
371      canClaimTokens = true;
372      vestingContract.setPrivatePhaseVestingStartTime(
373          block.timestamp + privatePhase.cliffDuration
374      );
375      vestingContract.setPublicPhaseVestingStartTime(block.timestamp + publicPhase.cliffDuration)
376      ;
376      emit ClaimingPhaseStarted(block.timestamp);
377  }
```

Listing 2.17: contracts/launchpad/LaunchPadV2.sol

```
178  /// @dev Returns whether the sale has already ended
179  function hasEnded() public view returns (bool) {
180      return publicPhase.endTime <= block.timestamp;
181  }
```

Listing 2.18: contracts/launchpad/LaunchPadV2.sol

Impact The claiming phase may be triggered at a time that does not align with the intended design, unexpectedly setting the start time for the vesting period.

Suggestion Add a check to ensure `publicPhase.endTime` is not equal to 0.

2.3 Additional Recommendation

2.3.1 Lack of check on `_startTime` in function `setPhase()`

Status Fixed in [Version 2](#)

Introduced by [Version 1](#)

Description In the function `setPhase()`, there is no check to ensure `privatePhase.startTime > block.timestamp`.

```
379 function setPhase(
380     uint32 _startTime,
381     uint32 _endTime,
382     uint256 _saleCap,
383     uint256 _tokenPerSaleToken,
384     uint256 _priorityMultiplier,
385     bool _isPrivate,
386     uint256 _cliffDuration
387 ) external onlyBeforeSale onlyOwner {
388     if (_startTime == 0 || _endTime <= _startTime || _endTime <= block.timestamp)
389         revert InvalidTime();
390     if (
391         _tokenPerSaleToken <= 0 ||
392         (_isPrivate &&
393             publicPhaseMaxTokenPerSale != 0 &&
394             publicPhaseMaxTokenPerSale > _tokenPerSaleToken)
395     ) revert InvalidPerSaleAmount();
396
397
398     PhaseInfo storage phase = _isPrivate ? privatePhase : publicPhase;
399
400
401     if (
402         (_isPrivate && publicPhase.endTime != 0 && _endTime != publicPhase.startTime) ||
403         (!_isPrivate && privatePhase.endTime != 0 && _startTime != privatePhase.endTime)
404     ) revert InvalidTime();
405
406
407     if (
408         (_isPrivate && publicPhase.saleCap != 0 && _saleCap > publicPhase.saleCap) ||
409         (!_isPrivate && privatePhase.saleCap != 0 && _saleCap < privatePhase.saleCap)
410     ) revert InvalidSaleCap();
411
412
413     phase.startTime = _startTime;
414     phase.endTime = _endTime;
415     phase.saleCap = _saleCap;
416     phase.tokenPerSaleToken = _tokenPerSaleToken;
417     phase.priorityMultiplier = _priorityMultiplier;
418     phase.cliffDuration = _cliffDuration;
419
420
421     emit PhaseUpdated(
```

```
422     _startTime,  
423     _endTime,  
424     _saleCap,  
425     _tokenPerSaleToken,  
426     _priorityMultiplier,  
427     _cliffDuration  
428 );  
429 }
```

Listing 2.19: contracts/launchpad/LaunchPadV2.sol

Suggestion Add relevant checks to ensure `privatePhase.startTime > block.timestamp`.

2.3.2 Incorrect check in function `getCurrentPhaseInfo()`

Status Confirmed

Introduced by Version 1

Description The view function `getCurrentPhaseInfo()` returns information about the current phase based on the current `block.timestamp`. When `block.timestamp` is exactly equal to `publicPhase.endTime`, this function considers the public phase is ongoing. However, the `hasEnded()` in the modifier `isSaleActive()` considers both sales, including the `public sale`, are inactive, which is inconsistent.

```
149  /// @dev Check whether the sale is currently active  
150  /// Will be marked as inactive if PROJECT_TOKEN has not been deposited into the contract  
151  modifier isSaleActive() {  
152      if (!hasStarted()) revert SaleNotStarted();  
153      if (hasEnded()) revert SaleCompleted();  
154      _;  
155  }
```

Listing 2.20: contracts/launchpad/LaunchPadV2.sol

```
178  /// @dev Returns whether the sale has already ended  
179  function hasEnded() public view returns (bool) {  
180      return publicPhase.endTime <= block.timestamp;  
181  }
```

Listing 2.21: contracts/launchpad/LaunchPadV2.sol

```
217  /// @dev Returns current running phase info.  
218  function getCurrentPhaseInfo()  
219  public  
220  view  
221  returns (bool isPrivatePhase, PhaseInfo memory phaseInfo)  
222  {  
223      uint256 currentBlockTimestamp = block.timestamp;  
224  
225  
226      if (  
227          currentBlockTimestamp < privatePhase.startTime ||  
228          currentBlockTimestamp > publicPhase.endTime
```

```
229     ) {
230         return (false, phaseInfo); // not started
231     }
232
233
234     if (currentBlockTimestamp < privatePhase.endTime) {
235         return (true, privatePhase);
236     }
237     return (false, publicPhase);
238 }
```

Listing 2.22: contracts/launchpad/LaunchPadV2.sol

Suggestion Revise the check to ensure that when `block.timestamp` is equal to `publicPhase.endTime`, the function returns the `phaseInfo` of the `public` phase.

2.4 Note

2.4.1 Potential centralization risks

Introduced by [Version 1](#)

Description The protocol includes several privileged functions, such as function `emergencyWithdrawFunds()`, `setCancellationFee()`. If the `owner`'s private key is lost or maliciously exploited, it could potentially cause losses to users.

